/*
 * Copyright (c) Meta Platforms, Inc. and affiliates.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

apply plugin: 'com.android.library'
apply plugin: 'kotlin-android'
apply plugin: 'com.facebook.kotlin.compilerplugins.dataclassgenerate'

project.group = GROUP
version = VERSION_NAME

android {
    compileSdkVersion rootProject.compileSdkVersion
    buildToolsVersion rootProject.buildToolsVersion

    useLibrary 'org.apache.http.legacy'

    kotlinOptions {
        freeCompilerArgs = ['-Xjvm-default=all']
    }

    defaultConfig {
        minSdkVersion rootProject.minSdkVersion
        targetSdkVersion rootProject.targetSdkVersion
    }

    buildFeatures {
        buildConfig true
    }

    testOptions {
        unitTests.includeAndroidResources = true

        unitTests.all {
            // Because of native libraries loading (Yoga), we can never reuse a class loader and
            // need to fork a new process per class.
            forkEvery = 1
            maxParallelForks = 4
        }
    }
    namespace 'com.facebook.rendercore'
    lint {
        abortOnError false
    }

    // TODO(#62): Re-enable abort on error.
}

dependencies {
  api project(":litho-rendercore-thread-utils")

  // Annotations
  compileOnly deps.jsr305
  implementation deps.inferAnnotations

  compileOnly deps.proguardAnnotations

  // Android Support Library
  compileOnly deps.supportAnnotations
  implementation deps.supportCore
  implementation deps.lifecycle

  testImplementation deps.assertjCore
  testImplementation deps.junit
  testImplementation deps.robolectric
  testImplementation deps.mockitoCore
  testImplementation deps.mockitokotlin
  testImplementation deps.supportAppCompat
  testImplementation deps.supportTestCore
  testImplementation deps.lifecycleService
  testImplementation project(":litho-rendercore-testing")
}

apply plugin: "com.vanniktech.maven.publish"
